
/* ACV_FNCT_dombox.sqf
----------------------------------------------------------
Crée une caisse de munition comme celle de la Domination.
On peut choisir le côté (side) et le format de la caisse.

[Position] call ACV_FNCT_DomBox;
[Position,"EAST"] call ACV_FNCT_DomBox;
[Position,"WEST","USOrdnanceBox"] call ACV_FNCT_DomBox;

@param1 Position
@param2 String (optionnel) Défaut : own side (Voir config de la mission)
@param3 class (optionnel) Défaut : air drop box (Voir config de la mission)

Note: d_own_side et d_enemy_side peuvent être utilisés pour le côté de la caisse.

Return : rien
-------------------------------------------------------*/
#include "ACV_constantes.h"
ACV_FNCT_DomBox = {
	private ["_args","_position","_side","_type"];
	_args = [];
	if (isNil "XfMissionCreateBox") exitWith {};
	if (typeName _this != "ARRAY") exitWith {};
	if (count _this <= 0) exitWith {};
	if (typeName (_this select 0) != "ARRAY") exitWith {};
	_position = _this select 0;
	If (count _position < 3) exitWith {};
	_args = _args + [_position];
	if (count _this > 1) then {
		If (typeName (_this select 1) == "STRING") then {
			_side = _this select 1;
			_args = _args + [_side];
			If (count _this > 2) then {
				If (typeName (_this select 2) == "STRING") then {
					_type = _this select 2;
					_args = _args + [_type];
				};
			};
		};
	};
	_args call XfMissionCreateBox;
};